
Transforming the Circuit 
 
While the circuit shown in Part 1, Figure 2 could be simulated directly, it is not optimal for SPICE.  
SPICE has some idiosyncrasies: 
 

o Current sources are better than voltage sources. 
 

o Laplace transform components work better if they converge to zero at high frequencies. 
 

o An inductor with an internal series resistance works better than a separate inductor and resistor. 
 

Currents are proportional to voltages. Voltage sources can be replaced with current sources. Voltage dependent sources 
can be replaced with current dependent sources. By applying some circuit theory transformations, the circuit in Figure 2 
can be converted to the circuit in Figure 3 by the successive application of equivalence transformations. This circuit looks 
greatly different from the circuit on Figure 2, but the terminal voltages and currents are the same.  

Figure 3. Two-port implemented with G components. The Laplace components are shown in yellow. 

 

There are 4 Laplace components: G1, G2, G7 and G8. G1 and G2 approach Z∞ instead of zero at high frequency. G7 and 
G8 do go to zero, but not quickly. Some improvement is needed. 

Figure 4. Unbalanced single mode transmission line in SPICE friendly form. The Laplace components are shown in yellow. 
 
Examination of the Laplace parameter of G1 and G2 reveals that it converges to a real constant 1/Z∞ which is a 
conductance. This constant will be subtracted out of the Laplace expression and added back as an ordinary resistor with 
impedance of Z∞. G1 and G2 will now converge to zero. To make G7 and G8 roll off more rapidly, the Laplace expression 
will be divided by (1 + s Lcon) which will cause the Laplace functions of G7 and G8 to converge quickly to zero. Then, to 
get the gain back, the output voltage of G7 and G8 is developed across an impedance equal to (1 + s Lcon).  
 
The result is shown in Figure 4. The 1-ohm resistor in series with Lcon will be absorbed into Lcon as a series resistance.  
The Tables' L1 and R1 will become the terminals of the sub-circuit. What value should be used for Lcon?  That will be 
discussed a little further down.  
 
The DC case 

Does the model work at DC?  It is not obvious from the unfamiliar forms that the solution takes. The characteristic 
impedance is very high. How, one might ask, can one get any signal into the transmission line at all?  The answer is that 
the line is mismatched at both ends and the reflections from the far end diminish the impedance at the near end and the 
reflections at the near end diminish the impedance at the far end. The two ends swap geometrically decreasing 



reflections that, with a little smoothing, look just like an RC-charging curve. The derivation and solution of the 
Telegrapher's Equations makes no assumption about the frequency of the signals and does not exclude DC. It ought to 
work. And in fact it does. 
 
How about SPICE?  Does it work with this model at DC?  The answer is yes for AC analysis, a qualified yes for transient 
analysis and a conditional yes for DC operating-point analysis. 
 
In the AC analysis, you cannot specify 0 Hz, but you can make the low frequency as low as you want (even one cycle per 
year ~ 31 nHz). The model works in the AC analysis with the understanding that the frequency must be positive. 
 
In the transient analysis, you put a step function (from zero to one, for example) and you see the waveform settles down 
to a sustained DC level. The model appears works in the transient analysis. 
 
What about DC operating point?  If the equations work and the model works and the transient and AC analysis work then 
the operating point analysis ought to work. Right?  Yes, it ought to work, but in some cases it does not. But it can be 
made to work, by increasing the inductance of Lcon.  
 
But, how do we feel about this?  We feel disquieted. We are of the opinion that this guiding principal should hold: if the 
model is a sufficiently accurate model of the physical system, then the simulator ought to produce accurate results. Thus, 
we should expend our effort on accurate modeling and not on appeasing the simulator. Once we get accurate simulation, 
we would be willing to tweak the model to run faster. 
 
We are not bothered by documented, universal principals such as current sources are preferred to voltage sources or that 
Laplace functions should go to zero for infinite frequency. We are not bothered by undocumented, universal principals 
that we have discovered, such as balanced circuits should be modeled by balanced models (which also is in accordance 
with our guiding principal of attempting to accurately model the physical system). We are a little bothered if we have to 
have a tweak, such as Lcon, that may need to be adjusted depending on rise times and transmission line length. But we 
are greatly bothered by having to change the value of the Lcon for different types of analysis. 
 
 
 



Choosing Lcon 

 
How can Lcon be determined?  We offer the following suggestions:   
 
Set up a simple operating-point circuit. Use a long (100 kft) transmission line. Drive it with a 1 Vdc source. Terminate it 
with Zdc. Run the operating point analysis. The output should be: 
 

8a.  
 
where 
 

8b.  
 

8c.  
 

8d.  
 
Reduce Lcon as small as possible and still get acceptably close. 
 
Set up a simple transient-analysis circuit with the length less than half of the shortest line you want to simulate. Match it 
at the source and leave it unterminated at the load. Drive it with a 1-volt step with a rise time about 1/10 of the 
propagation delay of the transmission line. Increase Lcon if needed to suppress anything weird. Weird things that we have 
seen include: 



 
 1. Bursts of large amplitude chaotic oscillation, usually after the simulation has converged to its final value.  
 2. Simulations that go on for a while outputting reasonable results, and which suddenly hang. 
 3. Cases where the delay was not proportional to the length of the transmission line. 
 4. Cases where the results have frequencies much higher than the input. 
 
You now have the minimum Lcon. Now increase Lcon until something changes significantly. If you can find a range of 
values for Lcon of, say, 10-to-one, where the outcome of a particular analysis does not change significantly, then set Lcon 
for the geometric mean of the range and use that value for that analysis. Certainly, when you do run any analysis you 
should try both twice and half the value for Lcon.  
 

Verifying the Simulation 
First, an AC simulation is run to see if the two-port accurately reproduces the data from Table 2.  
 

AC Simulation 

Figure 5. Frequency Response Schematic 



 

Figure 5 depicts the frequency response circuit. The SPICE program we use supports hierarchical schematics. Our 
brilliantly designed two-port has been reduced to a rectangle with two terminals. But this allows us to concentrate on the 
application.  
 
Table 2 gives loss in dB/Kft, so we will simulate a 1000-foot section. The propagation function is defined as the input-to-
output ratio when the transmission line is terminated by its characteristic impedance. We will use a longer transmission 
line of the same type for the termination.  
In the circuit of Figure 5, X1 is the 1000 foot section. X2 (1 million feet) serves as the load for x1.The results of the AC 
simulation are shown in Figure 6. The actual loss conformed exactly to the data in the Secondary Parameter 
Approximations columns of Table 2. For comparison, we run the same circuit using the SPICE O component with R, L and 
C from the highest frequency in Table 1. The O component is accurate for only one frequency. 
 

 
Figure 6. Matched Load Frequency Response Simulation: Telegrapher's Model in red and SPICE O component in green. 

 



Transient Simulation 

Figure 7. Transient Response Schematic. 

 
We will use the circuit of Figure 7 to try the Telegrapher's Model with transient simulation. The source, Ig, generates a 
Gaussian pulse of 20 ns width. We will try three cases: matched load, shorted load and open load. The circuit in Figure 7 
has a matched load. Actually Zinf is not a perfect match, except at infinite frequency, but it is close enough. 

Figure 8. Matched Load Transient Simulation. Input and output voltages are in blue and green respectively. 

 
Figure 8 shows the results for the matched load. The output pulse shows up about 147 ns after the input and there is no 
significant reflection. The output pulse is delayed, attenuated and slightly dispersed. 

Figure 9. Shorted Load Transient Simulation. Ro_1 is changed to 1μΩ. Ro_1 current shown in red. 

 

Figure 9 shows the results of a short-circuited load. We use 1 μΩ for a short circuit because it provides a component for 

measuring current. The current pulse shows up at about 147 ns after the input pulse and a reflected pulse of opposite 
polarity is seen at the input at the 294 ns later. The pulses show increasing dispersion as the travel time gets longer. 

Figure 10. Open Load Transient Simulation with step input. Ro_1 changed to 10 kΩ. 

 

Figure 10 shows the response to a one-volt step input on a line that is matched at the source and unterminated at the 
load. This is commonly called a "series termination" and is sometimes used for logic signals where there is a single 
source and a single load.  Even though the input takes two steps and 294 ns to reach 1 V, the output cleanly swings to 
one volt only 147 ns after the input. 



 
The sub-circuit 
The last thing we will do (and the first thing we promised to do) is to put all this circuitry into a sub-circuit that you can 
drop onto your SPICE schematic and get accurate time and/or frequency domain simulations. Figure 11 shows all of 
and nothing but the final sub-circuit. 

Figure 11. Unbalanced Transmission Line Sub-Circuit Schematic. 



 
The one-ohm resistors in series with L7 and L8 have been absorbed into those inductors as an internal series resistance. 
The Laplace expressions for G1, G2, g7 and G8 are shown at the bottom of the schematic. The chances of typing those 
expressions without error is low, so we are going to help you out with that. 
 
When SPICE was created, there were no graphical user interfaces (GUI's). Everything was entered as a text file using a 
plain old word processor or even punched cards. The syntax was standard and it was easy to swap files. Various vendors 
have added GUI's with non-standardized file structures. Each flavor of SPICE has its own unique method for creating sub-
circuits, creating symbol that represent the sub-circuit and passing parameters to those sub circuits. So, we cannot tell 
you how to create a sub-circuit on your version of SPICE. But we will help you. 
 
The old SPICE with its text files still exists underneath the GUI. Sub-circuit text files are still the most common way to 
provide SPICE models. Every SPICE GUI implementation has a way of accepting sub-circuit files as text and allowing its 
users to create symbols that link to those sub-circuit text files. If you request a SPICE model of an op amp or diode, you 
will get a plain old text file that you have to somehow wrap up into a graphical symbol. The odds are that if you have read 
all three parts of this article then you know how to do this already. So, open your word processor, go to the WWW version 
of this article, copy and paste the .param statement and the sub-circuit definition into your word processor and make the 
necessary manipulations. 
 
Or, if you want to create your sub-circuit graphically, we've done the hard part for you. Look at Figure 11 and draw it 
graphically. The hard part is the .param statement and the Laplace expressions. Once again the way to do that is to copy 
and paste it from the web version of this article. If you don't know how to do this, the Yahoo group listed as a resource is a 
good place to ask for help. 
 
Here is the sub-circuit net list. Note, in a .param statement, ** means exponentiation. In a Laplace expression ^ means 
exponentiation. The following text (shown in green courier font) is what you can copy and paste into your simulation. Once 
you get your sub-circuit entered, set the Kft parameter to the length of your transmission line and you are ready to 
simulate.  
 
 



.param Kft=1                ; 1 Kft = 1000 feet 

.param Lcon=10n             ; convergence inductance 

+ C=15.72e-9                ; the value of capacitance at dc  

+ Gdc=0.5n                  ; the value of conductance at dc  

+ Rdc=52.50                 ; the value of resistance at dc  

+ Ldc=0.1868e-3             ; the value of inductance at dc 

+ Linf=0.133e-3             ; inductance at infinite frequency 

+ Ldel=(Ldc-Linf)           ; inductance parameter 

+ Zinf=(Linf/C)**0.5        ; characteristic impedance at infinite  frequency 

+ Yinf=1/Zinf               ; characteristic conductance at infinite frequency  

+ F2=5e6                    ; the highest frequency in Hz 

+ W2=6.28318*F2             ; the highest frequency in rad/sec 

+ G1=23u                    ; the value of conductance at F1  

+ G2=36u                    ; the value of conductance at F2  

+ Rac=304.62                ; the value of resistance at F2  

+ F1=3e6                    ; the second highest frequency in Hz 

+ A=1.6                     ; inductance parameter 

+ k=Log(G2/G1)/Log(F2/F1)/2 ; conductance parameter 

+ WL=6.28318*161000         ; inductance parameter 

+ WR=W2*(Rdc**2)/(((Rac**4)-(Rdc**4))**0.5) ; resistance parameter  

 

.subckt single_mode_xline L1 R1 

G1 N1 0 N1 0 Laplace=(((((Gdc+G2*(-(s/w2)^2)^k)+s*C)/((Rdc*(1-(s/wR)^2)^0.25)+s*(Linf+Ldel/(1+A*((-(s/wL)^2)^0.5)-(s/wL)^2)^0.25)))^0.5))-Yinf 
G2 0 N1 N2 0 1 

G3 0 L1 N2 0 1 

V1 L1 N1 0 Rser=0 

H1 N4 0 V1 1 

G4 N6 0 N6 0 Laplace=(((((Gdc+G2*(-(s/w2)^2)^k)+s*C)/((Rdc*(1-(s/wR)^2)^0.25)+s*(Linf+Ldel/(1+A*((-(s/wL)^2)^0.5)-(s/wL)^2)^0.25)))^0.5))-Yinf 
G5 0 N6 N5 0 1 

G6 0 R1 N5 0 1 

V2 R1 N6 0 Rser=0 

H2 N3 0 V2 1 

R1 N6 0 {Zinf} 

R2 N1 0 {Zinf} 

G7 0 N2 N3 0 Laplace= Exp(-Kft*( (((Rdc*(1-(s/wR)^2)^.25)+s*(Linf+Ldel/(1+A*((-(s/wL)^2)^.5)-(s/wL)^2)^.25))*(Gdc+G2*(-(s/w2)^2)^k+s*C))^.5))/(s*Lcon+1) 

L1 N5 0 {Lcon} Rser=1 

G8 0 N5 N4 0 Laplace= Exp(-Kft*( (((Rdc*(1-(s/wR)^2)^.25)+s*(Linf+Ldel/(1+A*((-(s/wL)^2)^.5)-(s/wL)^2)^.25))*(Gdc+G2*(-(s/w2)^2)^k+s*C))^.5))/(s*Lcon+1) 

L2 N2 0 {Lcon} Rser=1 

.ends single_mode_xline 

Resources 
1. Linear Technology Corporation LT Spice IV download available with a very generous license. 
 http://www.linear.com/designtools/software/ltspice.jsp 
2. Yahoo user's group 
 http://tech.groups.yahoo.com/group/LTspice/ 

